

Set Theory or Higher Order Logic to Represent Auction Concepts in Isabelle?*

Marco B. Caminati¹, Manfred Kerber¹, Christoph Lange^{1,2}, and Colin Rowat³

¹ Computer Science, University of Birmingham, UK

² Fraunhofer IAIS and University of Bonn, Germany

³ Economics, University of Birmingham, UK

Project homepage: <http://cs.bham.ac.uk/research/projects/formare/>

Abstract When faced with the question of how to represent properties in a formal proof system any user has to make design decisions. We have proved three of the theorems from Maskin's 2004 survey article on Auction Theory using the Isabelle/HOL system, and we have verified software code that implements combinatorial Vickrey auctions. A fundamental question in this was how to represent some basic concepts: since set theory is available inside Isabelle/HOL, when introducing new definitions there is often the issue of balancing the amount of set-theoretical objects and of objects expressed using entities which are more typical of higher order logic such as functions or lists. Likewise, a user has often to answer the question whether to use a constructive or a non-constructive definition. Such decisions have consequences for the proof development and the usability of the formalization. For instance, sets are usually closer to the representation that economists would use and recognize, while the other objects are closer to the extraction of computational content. We have studied the advantages and disadvantages of these approaches, and their relationship, in the concrete application setting of auction theory. In addition, we present the corresponding Isabelle library of definitions and theorems, most prominently those dealing with relations and quotients.

1 Introduction

When representing mathematics in formal proof systems, alternative foundations can be used, with two important examples being set theory (e.g., Mizar takes this approach) and higher order logic (e.g., as in Isabelle/HOL). Another dimension in the representation is the difference between classical and constructive approaches. Again, there are systems which are predominantly classical (as most first order automated theorem provers) and constructive (e.g., Coq). Isabelle/HOL is flexible enough to enable the user to take these different approaches in the same system (e.g., although it is built on higher order logic, it contains a library for set theory, `Set.thy`). For instance, participants in an auction, i.e.

* This work has been supported by EPSRC grant EP/J007498/1 and an LMS Computer Science Small Grant. ...The final publication is available at <http://link.springer.com>.

bidders, can be represented by a predicate `Bidder x` or alternatively as a list of bidders `[b1, b2, b3]`.

The difference between a classical and a constructive definition can be demonstrated by the example of the argument of the maximum for a function (which we need for determining the winner of an auction). Classically we can define it, e.g., as $\text{arg_max } f \text{ } A = \{x \in A. f \ x = \text{Max}(f \ A)\}$. This definition is easy to understand but, unlike a constructive one, it does not tell how to compute `arg_max`. The constructive definition is more complicated (and has to consider different cases). It corresponds to a recursive function recurring on the elements of the domain. From a programming perspective, the two kinds of definitions just illustrated (classical versus constructive) can be seen respectively as *specification* versus *implementation*.

The approaches coexist in Isabelle/HOL. For example, for a set `X` one can apply the higher order function `f` by the construction `f ` X` to yield the image of the set. As a result, an author does not have to make a global decision of whether to use sets or higher order functions, but has fine grained control on what to use for a new mathematical object (e.g., sets vs. lists or lambda functions). Such a choice will typically depend on many factors. One factor is the task at hand (e.g., whether one needs to prove a theorem about an object or needs to compute its value). Another factor is naturalness of the constructions. This will typically depend on the authors and their expected audiences. Pragmatically, users also have to consider which of the possible approaches is more viable given the support provided by existing libraries for the system being used.

The ForMaRE project [8] applies formal methods to economics. One of the branches of economics the project focuses on is *auction theory*, which deals with the problem of allocating a set of resources among a set of participants while maximizing one or more parameters (e.g., revenue, or social welfare) in the process. ForMaRE has produced the Auction Theory Toolbox (ATT), containing Isabelle code for a range of auctions and theorems about them.⁴ Therefore, there is a good opportunity to practically test the feasibility of the different approaches, as introduced above, in this concrete setting. We adopted a pragmatic attitude: we typically took a set theoretic approach, since firstly we felt most familiar with it and secondly we knew from our ongoing interaction with economists that it would look more natural to them. However, when we needed to generate code, this generally excluded set-theoretical constructions such as the set comprehension notation. As a consequence most of our work was done in the set-theoretical realm, but some done constructively which allowed us to produce the code we wanted. The disadvantage of this approach is that we had to provide supplementary ‘compatibility’ proofs to show the equivalence of the set theoretic and the computable definitions when needed. This means that, as a byproduct of our efforts, we also generated a good amount of generic set-theoretical mater-

⁴ See <https://github.com/formare/auctions/>; the state at the time of this writing is archived at <https://github.com/formare/auctions/tree/1f1e7035da2543a0645b9c44a5276229a0aeb478>.

ial which was neither provided by the Isabelle library nor Isabelle’s Archive of Formal Proofs (AFP)[6].

In this paper, rather than discussing our progress in the application domain (auction theory), we illustrate this material and its relationship to the mathematical objects we needed. We will discuss this by the following three concrete examples that occurred while we were working on the ATT:

1. We model a function indirectly through the set-theoretical notion of the graph of a relation, rather than directly using the HOL primitive notion of a lambda-abstracted function. This allowed us to concisely define auction-related notions through two natural set-theoretical constructions (extension and restriction of a function). Moreover, this in turn allowed us to generalize some theorems from functions to relations. We also discuss how this choice does not necessarily mean giving up the advantage of computability (section 2). Incidentally, this will also give insight into the main differences between set theory as implemented in Isabelle/HOL and standard set theory.
2. We developed a stand-alone formalization of the definition of functions over equivalence classes (section 3). This is a common construction when, e.g., defining the canonically induced operation on the quotient of a group by one of its normal subgroups. To define such an operation, we proved the invariance (or well-definedness) under an equivalence relation.
3. We applied a set-theoretical, non-constructive definition and a constructive one to the same object (the set of all possible partitions of a set). The two approaches were used for different purposes (theorem proving and computation, respectively) and proved formally their equivalence by a ‘compatibility’ theorem (section 4). The same approach is adopted for the set of all possible injective functions between sets.

In existing proof assistants based on set theory (e.g., Mizar, Metamath), the desirable quality of being able to compute values of functions (rather than only the truth value of predicates involving them), is typically lost⁵.

We will present examples which show that we can retain this Isabelle/HOL-induced advantage in our developments.

Overview of the paper

Section 2 introduces and motivates the set-theoretical encoding of functions, as an alternative to a lambda representation; this is done by commenting on the relevant Isabelle definitions, and by illustrating a particular theorem regarding auctions, which employs those definitions. Section 3 brings this approach to a more abstract level, by showing how it can handle, given an initial function, the

⁵ We do not know if this is due to how the existing proof assistants are implemented, or to some fundamental limitation of set theory. Reading the recent “Computational set theory” thread (<http://www.cs.nyu.edu/pipermail/fom/2014-February/017841.html>), and its related threads, on the “Foundations of Mathematics” mailing list, we feel the question is currently open.

definition of a second function on the corresponding equivalence classes. Since this was already done in Isabelle’s `Equiv_Relations.thy`, we also illustrate the differences with it.

In some cases, defining a mathematical object in purely set-theoretical terms does not preserve computability of that object: section 4 introduces a technique for such cases. As mentioned in point 3 above, this technique introduces a parallel, computable definition, and then formally proves the equivalence of the two definitions. In the same section, this technique is presented through two examples from our ATT; we then discuss how we took advantage from having preserved both the definitions. Finally, section 5 explains how we applied the general machinery from section 3 to the ATT.

2 Set-theoretical Definition of Functions in Isabelle/HOL

In higher order logic, functions, function abstraction, and function application are primitives [3]; in set theory, the primitive notion is that of a set, and a function f is represented by its graph, i.e., the *set* of all the ordered pairs $(x, f(x))$. We chose to work mostly with the set-theoretical representation of the functions even though we are using a tool based on higher order logic. In the following we give reasons why.

1. A first reason is that a set theoretical representation more easily allows to enumerate all the (injective) functions from a finite set to another finite set ([4]). In contrast, this seems to be more complicated to do directly in higher order logic, where all the functions are assumed to be total.
2. A second reason is that the set-theoretical, graph-based representation works even for generic relations, thus often allowing us to extend the results we proved to the more general situations involving relations, rather than functions (see the next subsection).
3. A third reason is that the operation of function restriction is naturally expressed in terms of two elementary set operations: cartesian product and intersection. Indeed, this allows to extend this operation to relations, immediately giving an instance of what we argued in point (2) above. Restriction is a fundamental operation for representing the concept of weakly dominant strategy, a key concept in auction theory, see e.g., [10, proposition 2]. The definition of function restriction is arguably more complicated in standard higher order logic since functions are always total. That is, restricting them to a set requires carrying the restricted domain set with the function, e.g. by forming a pair $(R, f|_R)$, whereas the set-theoretical representation naturally includes this set.
4. Finally, specific partial, finite set theoretical functions can be very concisely and quickly defined; this made it possible to promptly test Isabelle code while we were working on it. For example, functions can be written in the form of a set as $\{(0,10), (1,11), (2,12)\}$ and fed to the Isabelle definitions very easily in order to test the correctness of related computations empirically, whereas a lambda expression would be more complex to define.

2.1 Two basic operators on relations: ‘outside’ and pasting

In this section, we discuss two general mathematical operations we encountered specifically during formalization of auctions. Assume a set of bidders N and a function $b : N \rightarrow \mathbb{R}$ that determines the corresponding bids. The first operation removes one bidder i from the domain N of the bid function. The second operation alters the bid function in one point i of its domain, $b +*(i, b_i)$, which is equal to b except for argument i , where the value is changed to b_i . In set theory, a function (or a relation) inherently specifies its domain and range. A generalization of the first operation is thus obtained by writing (in Isabelle):

```
definition Outside ::  
  "('a × 'b) set ⇒ 'a set ⇒ ('a × 'b) set"  
  (infix "outside" 75) where  
  "R outside X = R - (X × Range R)",
```

where 75 denotes the binding strength of the infix operator. The following specialization to singletons ...

```
abbreviation singleoutside (infix "--" 75) where  
  "b -- i ≡ b outside {i}",
```

... turned out handy for our purposes.

Another circumstance making the set theoretical approach convenient is that now the second operation can be obtained in simple terms of the first as follows:

```
definition paste (infix "+*" 75)  
  where "P +* Q = (P outside Domain Q) ∪ Q",
```

which can be specialized to the important case of Q being a singleton function:

```
abbreviation singlepaste where  
  "singlepaste F f ≡ F +* {(fst f, snd f)}"  
  notation singlepaste (infix "+<" 75).
```

While we often applied `outside` and `+*` to functions rather than relations (i.e., assuming right uniqueness, see section 2.2), many of its properties were proved for relations in general. For example, the associativity theorem

```
lemma 1153: "(P +* Q) +* R = P +* (Q +* R)"
```

has no hypotheses in its statement and holds for general relations.

2.2 Specializing relations to functions: right-uniqueness and evaluation

The operators `outside` and `+*` are building blocks on which the statements of many theorems we proved for auctions are based. In turn, a number of preparatory lemmas have been proven about those objects in the generic case of relations; however, others only hold when considering relations which actually are functions. Hence, we need the following predicate for *right-uniqueness* of a relation, which we define in terms of the *triviality* of a set (i.e. being empty or singleton):

```
definition trivial where "trivial x = (x ⊆ {the_elem x})",
```

where `the_elem` extracts an element from a set, being undefined when this cannot be done in a unique way. The predicate for right-uniqueness is called `runiq`, and it uses the operator `R ‘ X`, which yields the image of the set `X` through the relation `R`:

```
definition runiq :: "('a × 'b) set ⇒ bool" where
"runiq R = (∀ X . trivial X → trivial (R ‘ X))".
```

We note that, contrary to other proof assistants based on set-theory (e.g., Mizar), which in general cannot directly compute values of functions (at most the truth or falsity of predicates involving those values), we are able to preserve from Isabelle/HOL the ability of actually computing the evaluation of these set-theoretical flavoured functions, when they are right-unique, through the following operator:

```
fun eval_rel :: "('a × 'b) set ⇒ 'a ⇒ 'b" (infix " ,," 75)
where "R ,, a = the_elem (R ‘ {a})"
```

Now, indeed, set theoretical functions can be evaluated via `,,`. This can be tested through the Isabelle command `value`: for example we can write

```
value "{(0::nat,10),(1,11),(1,12::nat)} ,, 0"
```

and obtain 10 as an answer. This holds also when combining `eval_rel` with the operators as from the beginning of this section; for example

```
value "({(0::nat,10),(1,11),(1,12)} +< (1,13::nat)) ,, 1"
```

yields the answer 13.

A right-unique relation and a standard higher order logic, lambda abstracted function represent the same mathematical object, hence it should be possible to pass from one representation to another. `graph` from `Function_Order.thy`, defined as

```
definition graph where "graph X f = {(x, f x) | x. x ∈ X}"
```

does exactly that. The opposite conversion can be achieved easily as follows:

```
definition toFunction (* inverts graph *)
where "toFunction R = (λ x . (R ,, x))"
```

However, the degree of computability of set-theoretical functions is less than with original HOL functions; for example we cannot evaluate

```
value "(graph {x::nat. x<3} (λx. (10::nat))),,(1::nat)",
```

while the following works as expected

```
value "(graph {0,1,2} (λx. (10::nat))),,(1::nat)".
```

We also note that, since Isabelle formalizes set theory inside higher order logic, types still impose some rigidity, compared to stand-alone set theory: see section 6. For example, the following alternative Isabelle definition would be exactly equivalent to `eval_rel` in a standard (untyped) set theory:

```
abbreviation "eval_rel2 (R::('a×('b set)) set) (x::'a)
≡ ⋃ (R '{x})" notation eval_rel2 (infix ".,," 75),
```

It is, however, actually defined (and equivalent to it) *only* for set-yielding relations:

```
lemma 11182: assumes "runiq (f::('a × ('b set)) set)"
"(x::'a) ∈ Domain f" shows "f.,x = f.,.,x"
```

However, when it is applicable, `eval_rel2` has the desirable qualities of evaluating to the empty set outside the domain of `f`, and in general to something defined when right-uniqueness does not hold (in which case `eval_rel` is undefined). This allowed us to give more concise proofs in such cases.

`runiq` is a central definition in our formalization; its many possible equivalent formulations have turned out to be useful in different steps when proving various lemmas. Here we present the possible alternative definitions we have proven to be equivalent in the ATT:

```
lemma 11133: "runiq P=inj_on fst P"
lemma runiq_alt: "runiq R ↔ (∀ x . trivial (R '{x}))"
lemma runiq_basic:
"runiq R ↔ (∀ x y y' . (x, y) ∈ R ∧ (x, y') ∈ R → y = y')"
lemma runiq_wrt_eval_rel:
"runiq R ↔ (∀x . R '{x} ⊆ {R ., x})"
lemma runiq_wrt_eval_rel':
"runiq R ↔ (∀x ∈ Domain R . R '{x} = {R ., x})"
lemma runiq_wrt_ex1:
"runiq R ↔ (∀ a ∈ Domain R . ∃! b . (a, b) ∈ R)"
lemma runiq_wrt_THE:
"runiq R ↔ (∀ a b . (a, b) ∈ R → b = (THE b . (a, b) ∈ R))"
```

In general, we found that, especially for basic and ubiquitous concepts such as `runiq`, the more equivalent definitions we have, the better. One reason is that this improves the understandability of the formalization: different readers will find different definitions easier to grasp. Another reason is that automated theorem proving tools, such as Sledgehammer⁶, will be more likely to find automated justification in single steps of subsequent proofs: by picking the appropriate equivalent definition, sledgehammer can find a justification, while, upon removing that definition, it is no longer able to do that. We actually experienced this phenomenon with proofs involving `runiq`: the form of `runiq` given in 11133 allowed Sledgehammer to find the proof of this technical lemma:

```
lemma 11134: assumes "runiq P" shows "card (Domain P) = card P".
```

Lemma 11134 above was in turn used to formalize proposition 3 from [10].

⁶ Sledgehammer is an Isabelle tool that applies automatic theorem provers (ATPs) and satisfiability-modulo-theories (SMT) solvers to automatically produce proofs [2].

2.3 Application to auctions

Next, we give one example of the roles of the operations introduced in this section in our practical setting of auctions, for the simple case of a single-good auction. In this case, the input data for the auction are given through a function \mathbf{b} associating to each bidder the amount she bids for the good. Given a fixed bidder i , the outcome of the auction is determined by two functions, \mathbf{a} and \mathbf{p} . Both take \mathbf{b} as an argument: the first yields whether that bidder won the item ($\mathbf{a}, \mathbf{b} = 1$) or not ($\mathbf{a}, \mathbf{b} = 0$); the second, \mathbf{p}, \mathbf{b} , yields how much she has to pay. We take the Isabelle formalization of the second proposition in [10], which is theorem `th10` in file `Maskin2.thy`. This result proves, given some general requirements, the logical equivalence of two properties, each binding \mathbf{a} and \mathbf{p} :

- The first property we called `gen Vick` (for generalized Vickrey auction, see below); it states that the payment imposed to i is the sum of a ‘fee’ term $\mathbf{t}(\mathbf{b}--i)$, to be paid irrespectively of the outcome, and of a proper price term $(\mathbf{a}, \mathbf{b} - \mathbf{a}1)*\mathbf{w}(\mathbf{b}--i)$, which is to be added only in case she obtains the good. Moreover, the first term does not depend on how much i herself bids, but only on others’ bids. The proper price is determined by the auxiliary function \mathbf{w} , which also does not depend on i ’s bid (since Vickrey auctions are second price auctions). Hence, i ’s bid can influence only whether i pays or not the proper price, but not its amount.
- The second property, called `dom4`, states that i can never be worse off if she changes her bid to her real valuation v of the good.

Hence, `gen Vick` assumes the following form:

```
abbreviation gen Vick where "gen Vick a p i w ≡
(∃ (a1::allocation) t. (∀ b ∈ Domain a ∩ (Domain p).
p, b = (a, b - a1)*w(b--i) + t(b--i) ))",
```

while `dom4` is the following inequality:

```
definition dom4 where "dom4 i a p = (∀ b::bid. ∀ v.(
{b, b+<(i,v)} ⊆ (Domain a ∩ (Domain p)) ∧ i ∈ Domain b) →
v*(a, b) - (p, b) ≤ v*(a, (b+<(i,v))) - (p, (b+<(i,v)))")".
```

The operators `--` and `+<` are central in expressing those two conditions; their respective general properties (collected in `RelationProperties.thy`) permitted to streamline the proof of the theorem, whose thesis reads:

$$\text{gen Vick } a \ p \ i \ w = \text{dom4 } i \ a \ p. \tag{1}$$

3 Quotients between relations

We built a library of basic facts centred around our new constructs of right-uniqueness (`runiq`), evaluation $(,)$, pasting $(+*)$, and considering a function outside some subset of its domain. The library also contains more advanced results. In particular, we describe here our approach to building quotients; then,

we derive functions on equivalence classes of points from functions defined on single points. These methods are common in many areas of mathematics, especially algebra and topology; they are used when a given property holds on classes of objects, and one wants to abstract away from the specific representative of a class, and rather define the given property on the whole class. Such classes are typically the equivalence classes induced by an equivalence relation over its domain, which form the *quotient* of the original set.

For example, in group theory, the operation of a group G is canonically transported to the set of the cosets yielded by a normal subgroup N . This is what makes the quotient group G/N a group, and is only possible if the group operation is *class-invariant*⁷: the product of two representative elements of two cosets must be in the same coset, irrespective of how those representative elements are selected. This is ensured by the definition of normal subgroup.

In our case, to construct `t` appearing in the definition of `genvick` towards the end of the previous section, we need to define some function taking as an argument a bid vector with the i -th component removed (where i is a bidder). The formal way to do that was to aggregate all possible bid vectors differing only on their i -th component and to define that function on the classes obtained this way; hence, using quotients naturally emerged as one elegant approach. More details on this particular application of quotients are in section 5.

As illustrated in [12], existing Isabelle’s theory `Equiv_Relations.thy` already introduces tools for these general mathematical techniques. There are, however, the following problems:

1. The operation of passing from a pointwise function to the ‘abstracted’ version defined on equivalence classes is done using type-theoretical `Abs_` and `Rep_`. In paper-based mathematics, on the other hand, everything is done using set theory; hence, a mathematician would probably not know how to use this implementation without first getting some knowledge of the underlying type-theoretical foundations.
2. This operation must be performed ‘manually’ in each separate case: there is no generic definition to do that given a pair (f, R) , where f is the function to be abstracted and R is an equivalence relation on its domain. In contrast, in our treatment we introduce a function called `quotient` doing exactly this.
3. f and R are typed as a lambda function and a set-theoretical relation, respectively, while there is no reason to preemptively preventing f from being a generic relation (i.e., not necessarily being right unique).

For these reasons, we coded a purely set-theoretical Isabelle implementation of this machinery, via three simple definitions and one theorem establishing the right-uniqueness of the abstracted function given basic requirements on the pointwise function.

The first definition gives a map to pass from a point of the domain of a relation `R` to the corresponding equivalence class:

⁷ In this case we can also say that the group operation is *well-defined*, or that it *respects* the corresponding equivalence relation, or even that it is *compatible* with it.

```

definition projector where
  "projector R = {(x,R' '{x}) | x. x ∈ Domain R }"

```

The second definition builds, given a pointwise relation R and two equivalence relations P, Q (working on its domain and codomain, respectively) the corresponding, abstracted relation on the resulting equivalence classes:

```

definition quotient where "quotient R P Q =
  {(p,q) | p q. q ∈ (Range (projector Q)) ∧
  p ∈ Range (projector P) ∧ p × q ∩ R ≠ {} }"

```

While this definition is typically given for a function R and equivalence relations P and Q, it still makes sense if these additional conditions are not satisfied. This allows us to lift these requirements in some preparatory lemmas before assuming them in the following main result:

```

lemma 123: assumes "compatible f P Q" "runiq f" "trans P" "sym P"
  "equiv (Domain Q) Q" shows "runiq (quotient f P Q)",

```

where the predicates `equiv X P`, `trans P`, `sym P` exist in the Isabelle library, and state, respectively, that P is an equivalence relation over the set X; that P is a transitive relation; and that P is a symmetric relation.

Note that in Isabelle there is a definition for a quotient of a relation R written as `quotient R` as the set of all equivalence classes associated with R. Here, however, we assume a function (or relation) `f` with respect to relations P and Q and define the quotient of `f` with respect to R as a function with the domain `quotient R`.


The notion of compatibility above asks that `f` respects P and Q:

```

definition compatible where
  "compatible R P Q =
  (∀ x . (R' '(P' '{x}) ⊆ Q' '(R' '{x})))"

```

Note that the definition of `compatible` is a generalization of the usual commutativity of the application of functions as displayed to the right.



It should be noted that this condition is not required at the time of defining a quotient, which can be defined for any triple of relations; rather, it is only required when asking that the quotient behaves in the expected way, as stated by `lemma 123` above. This allows us to freely use the construction `quotient` in advance, to show a number of intermediate results not requiring compatibility themselves. For example, the following lemma holds for general relations:

```

lemma quotientFactors: assumes "equiv (Domain p) p" "equiv (Domain q) q"
  shows "quotient r p q = (projector p)- ∘ r ∘ (projector q)".

```

where `R-` is the converse of the relation R and `∘` stands for relation composition.

We also note that, for similar reasons, we devised a definition of compatibility taking any triple of relations as arguments: as in the definition of `quotient`, one can use `compatible R P Q` even before showing that R is right-unique and that P, Q are equivalence relations. In the latter case, however, the definition of compatibility reduces to asking that any two Q-equivalent points have P-equivalent images through R.

4 Injective functions and partitions

In section 2.3, we introduced the mathematical description of a single-good auction. An important family of more complex schemes is given by *combinatorial* auctions. In these, there are several objects at stake (a set of goods G), and each bidder (of a set N) can bid for each possible combination of them. The outcome of the auction is still described by a pair of maps (a, p) , yielding respectively what a bidder gets and how much she has to pay. However, now the “what a bidder gets” part must be represented by a mathematical object more articulate than a $\{0, 1\}$ -valued function. It is represented by a partition of G , and by an injective function (or injection) from that partition to N . Treating the latter function inside set-theory gives an advantage: the pair (partition, injection) is conveniently represented by the injection alone, because the partition of G will be simply the domain of the former. This allowed us to type the relevant objects plainly as follows:

```
type_synonym bidder = "nat"
type_synonym goods = "nat set"
type_synonym allocation_rel = "(goods × bidder) set".
```

Since we wanted to extract code from our Isabelle formalization, we had to implement recursive definitions for both the set of all possible partitions of a finite set and for the set of all possible injections from a finite set to another finite set. In both cases, those recursive definitions turned out to be inconvenient when it came to prove mathematical facts involving them. Hence a separate, more natural definition was needed that is equivalent but not computable.

We illustrate this dual approach in the case of injections.

```
fun injections_alg ::
  "'a list ⇒ 'b::linorder set ⇒ ('a × 'b) set list" where
  "injections_alg [] Y = [{}]" |
  "injections_alg (x # xs) Y = concat
  [[R**{(x,y)}. y←sorted_list_of_set (Y-Range R)].
   R ← injections_alg xs Y]"
```

Table 1. Constructive definition of injection

In Table 1, we find a definition of all injective functions between two finite sets X and Y , which recurs on X , while in Table 2 the set of all injective functions is defined axiomatically.

```
definition injections :: "'a set ⇒ 'b set ⇒ ('a × 'b) set set"
where "injections X Y =
  {R. Domain R=X ∧ Range R⊆Y ∧ runiq R ∧ runiq(R-)}"
```

Table 2. Axiomatic definition of injection

We use the constructive definition in computations and the axiomatic for proofs and mathematical manipulations. In order to do that we have to prove their equivalence in the theorem stated in Table 3.

```

theorem injections_equiv:
fixes xs::" 'a list"
and Y::" 'b::linorder set"
assumes non_empty: "card Y > 0"
shows "distinct xs  $\implies$ 
(set(injections_alg xs Y)::('a $\times$ 'b)set set)=injections (set xs) Y"

```

Table 3. Compatibility of constructive and axiomatic definitions of injection

Similarly, we have a constructive and an axiomatic definition for partitions (see Tables 4 and 5, respectively).

```

definition insert_into_member_list
:: " 'a  $\implies$  'a set list  $\implies$  'a set  $\implies$  'a set list"
where "insert_into_member_list new_el Sets S =
(S  $\cup$  { new_el }) # (remove1 S Sets)"

definition coarser_partitions_with_list
::" 'a  $\implies$  'a set list  $\implies$  'a set list list" where
"coarser_partitions_with_list new_el P = ({ new_el } # P)
#
(map ((insert_into_member_list new_el P)) P)"

definition all_coarser_partitions_with_list
::" 'a  $\implies$  'a set list list  $\implies$  'a set list list" where
"all_coarser_partitions_with_list elem Ps =
concat (map (coarser_partitions_with_list elem) Ps)"

fun all_partitions_list :: " 'a list  $\implies$  'a set list list" where
"all_partitions_list [] = [[]]" |
"all_partitions_list (e # X) =
all_coarser_partitions_with_list e (all_partitions_list X)"

```

Table 4. Constructive definition of partition

The constructive definition above represents a partition of a finite set as a list of (disjoint) subsets of it; it works by induction on the cardinality of the set to be partitioned, as follows. Given a set of cardinality $n + 1$, we write it as a disjoint union $X \cup \{x\}$, and, given a partition P of X , we insert x into each set belonging to P (this is done by the operator `insert_into_member_list` above). We thus obtain $|P|$ many partitions of $X \cup \{x\}$, to which we add the distinct partition $\{\{x\}\} \cup P$. Making P range over all possible partitions of X , we obtain in this way all possible partitions of $X \cup \{x\}$.

When implementing this idea into the code above, however, we chose to represent the finite set X as a list (rather than a set) of elements of it, and a partition of it as a list (rather than a set) of subsets of it. This is because the algorithm just described is iterated in two ways: first, x is inserted into each set of a partition; secondly, the construction is iterated for each possible partition of X . Since iterations are easier with lists than with finite sets [11], we adopted that choice.

```

definition is_partition where
"is_partition P = ( $\forall X \in P . \forall Y \in P . (X \cap Y \neq \{\} \leftrightarrow X = Y)$ )"

definition is_partition_of (infix "partitions" 75)
where "is_partition_of P A = ( $\bigcup P = A \wedge \text{is\_partition } P$ )"

definition all_partitions where
"all_partitions A = {P . P partitions A}."

```

Table 5. Axiomatic definition of partition

It should be noted that passing from a list to a finite set (via the operator `set`) is easier than the converse, hence the equivalence theorem for partitions in Table 6 is stated taking as input data a list `xs`:

```

theorem all_partitions_paper_equiv_alg:
  fixes xs::"a list"
  shows "distinct xs  $\implies$ 
  set(map set (all_partitions_list xs)) = all_partitions(set xs)"

```

Table 6. Compatibility of constructive and axiomatic definitions of partition

Thanks to the constructive definitions above, we were able to extract, from Isabelle code [5], executable code running Vickrey combinatorial auctions [4]. Furthermore, we were able to prove fundamental theoretical properties about those auctions by using the non-constructive versions of those definitions; for example, that the price for any bidder is non-negative. The compatibility theorems in Tables 3 and 6 allow to certify that such theoretical properties hold also for the extracted code.

We note that, while non-constructive versions hold in the general case, constructive versions are obviously limited to the finite case (e.g., calculating all the partitions of a finite set). Therefore, the two compatibility theorems must restrict to the finite case: this is reflected by the fact that the argument of `injections` and `all_partitions`, appearing in Tables 3 and 6, respectively, is `set xs`. This means that such an argument is automatically a finite set, being the result of converting a list (`xs`) to a set using the Isabelle function `set`. We stress the fact that our approach allows to prove most theorems (e.g., thesis (1) at the end of section 2) without restricting to the finite case, and to add the additional hypothesis of finiteness only and exactly for the theorems needing it.

5 Application of quotients to auctions

We now give enough formalization details to illustrate the exact point in our proofs of auction theory in which we needed to employ `projector` and `quotient`, introduced in section 3.

To prove the thesis (1) at the end of section 2, we needed to explicitly build the `t` appearing in the definition of `genvick`. This function is uniquely determined by `a` and `p`; however, the latter takes `b` as an argument, while `t` takes `b--i` (which represents a bid vector with bidder `i`'s bid removed, and is called a *reduced bid*). The algebraic way to pass from `b` to `b--i` is to consider an equivalence

relation associating any two b b' differing at most in the point i . Correspondingly, to obtain from p a function on arguments of the form $b--i$, we form the quotient of p according to that equivalence relation. The equivalence relation we need is exactly the kernel⁸ of the following function:

```
definition reducedbid:: "bidder  $\Rightarrow$  (bid  $\times$  allocation) set  $\Rightarrow$ 
(bid  $\times$  bidder set  $\times$  bid  $\times$  allocation) set"
where "reducedbid i a =
{(b, (Domain b, b outside {i}, a ,, b)) | b. b  $\in$  Domain a}".
```

So that the explicit construction of t , used inside the proof for (1), ends up as

```
" $\lambda$ x. reducedprice p i a ,, ({i}  $\cup$  Domain x, x, Min (Range a))",
```

where

```
definition reducedprice:: "(bid  $\times$  price) set  $\Rightarrow$  bidder  $\Rightarrow$ 
(bid  $\times$  allocation ) set  $\Rightarrow$ 
((bidder set  $\times$  bid  $\times$  allocation)  $\times$  price) set"
where "reducedprice p i a =
(projector ((reducedbid i a)-)) 0
(quotient p (Kernel (reducedbid i a)) Id) 0
((projector Id)-)",
```

Id being the identity function. With this definition in place, an important part of the theorem consists in showing that `reducedprice` is right-unique.

This reduces to showing that `quotient p (Kernel(reducedbid i a)) Id` appearing in the definition above is right-unique. Thanks to `lm23` (see section 3), this in turn means proving the compatibility between p and the equivalence relation just introduced, i.e., `Kernel (reducedbid i a)`, which is provided by

```
lemma l24b: assumes "functional (Domain a)" "Domain a  $\subseteq$  Domain p"
"dom4 i a p" "runiq p" shows
"compatible p (Kernel (reducedbid i a)) Id".
```

6 Discussion and related work

The work presented here is based on the specific set theory implemented in Isabelle/HOL, sometimes called simply-typed set theory [14, Section 6.1], [13, Section 1]. It differs from standard set theories, as Zermelo-Fraenkel (ZF), in that the primitives of the latter are encoded using primitives of higher order logic, as follows: a given set p is actually a term p of type $\tau \Rightarrow \text{bool}$, and the writing $x \in p$ is actually the application $p x$. This means that in Isabelle/HOL one cannot write a set like $\{x, \{x\}\}$ (which is not well-typed): the standard hierarchy of ZF is no longer constructable. As long as the objective is to formalize ordinary (as defined in [15, Section I.1]) mathematics, this usually causes no problem: while this difference prevents the encoding of relevant mathematical objects

⁸ We recall that the kernel of a function f is the equivalence relation \circ_f given by $x_1 \circ_f x_2 \iff f(x_1) = f(x_2)$. See [1, Definition 1.18]. The kernel notion was missing in the Isabelle library, and we also provided it in ours.

(e.g., natural numbers, integers, reals, cartesian products) as usually done in ZF, those objects can be directly represented in higher order logic. This means that, for many mathematical branches, what is lost with respect to ZF is limited to its more technical side-effects, as $\pi \cap \mathbb{Q} \neq \emptyset$, or $1 \cap 2 \subseteq 3$. Since the latter are often regarded as strange or meaningless writings [9], [13, Section 1], this could even be a desirable consequence.

On the other hand, when trying to exploit technical ZF ‘hacks’, as we tried to do with `eval_re12`, problems can arise: we discussed that in section 2.2. Moreover, ZF is a more appropriate tool for studying the remaining branches of mathematics, starting with set theory itself.

The Z pattern catalogue [16, Section IV] allows to represent functions as sets of pairs, as done here. However, Z is a specification language, while we are interested in both specification and implementation. Mizar is based on untyped set theory, thus modelling functions and relations as done here, and also provides many relevant existing theorems in its library; however, it is not possible to extract code or to do computations in general. The Ssreflect extension library for the Coq proof assistant is extensive and provides a lot of operations; on the other hand, most of this material seems to apply to functions as represented by lists, thereby inherently limiting to the finite case. More details on how Isabelle/HOL generally compares with other systems are in the comparative study [7].

7 Conclusions

We have built an extensive library of results about functions, represented as right-unique relations as an alternative to the lambda abstractions that have so far been typical of Isabelle/HOL. In this paper we explained how we employed this alternative technique, and the concepts in our library, in the application domain of auction theory. Our library ranges from simpler constructions such as passing a relation onto another, to more sophisticated ones such as quotients.

We described how set-theoretical constructs can concisely express constructions that frequently occur in formalization, giving concrete examples from the application domain of auction theory. In the particular case of representing functions, we discussed the advantages and disadvantages of the set-theoretical representations with respect to the alternative, more natural (given the foundations of Isabelle/HOL) lambda representation. We also showed that computability, a typical feature of lambda functions, can be achieved for set-theoretical functions in Isabelle/HOL. Moreover, even in those cases in which it cannot be achieved, we showed how we were still able to employ a dual approach, giving non-constructive, more expressive definitions along with constructive (and thus computable) ones, finally proving their equivalence through compatibility theorems. This allows us to obtain the best of both worlds, that is, expressiveness and proof-friendly definitions together with computability, at the cost of having to prove the additional compatibility theorems.

Finally, we took our application of this approach to a more advanced, abstract level by describing in a simple, close-to-paper set-theoretical style notions such as

quotients, compatibility and *kernel*. We presented a concrete application of this material in a hands-on case encountered in formalizing auction theory within the ForMaRE project. We think that our library is rich and generic enough to be of possible use to other Isabelle users, and we hope that this paper can serve as a first guiding example to show how it can be employed.

References

- [1] C. Bergman. *Universal Algebra: Fundamentals and Selected Topics*. Chapman & Hall Pure and Applied Mathematics. Taylor & Francis, 2011.
- [2] J. C. Blanchette and L. C. Paulson. *Hammering Away. A User's Guide to Sledgehammer for Isabelle/HOL*. 5th Dec. 2013.
- [3] J. Bowen and M. Gordon. 'Z and HOL'. In: *Z User Workshop, Cambridge 1994*. Springer, 1994, pp. 141–167.
- [4] M. B. Caminati et al. *Proving soundness of combinatorial Vickrey auctions and generating verified executable code*. 2013. arXiv:1308.1779 [cs.GT].
- [5] F. Haftmann and T. Nipkow. 'Code generation via higher-order rewrite systems'. In: *Functional and Logic Programming*. Springer, 2010, pp. 103–117.
- [6] G. Klein et al., eds. *Archive of Formal Proofs*. 2014.
- [7] C. Lange et al. 'A qualitative comparison of the suitability of four theorem provers for basic auction theory'. In: *Intelligent Computer Mathematics: MKM, Calculemus, DML, and Systems and Projects 2013*. Ed. by J. Carrette et al. Lecture Notes in Artificial Intelligence 7961. Springer-Verlag, 2013, pp. 200–215.
- [8] C. Lange et al. 'The ForMaRE Project – Formal Mathematical Reasoning in Economics'. In: *Intelligent Computer Mathematics*. Ed. by J. Carrette et al. LNCS 7961. Springer, 2013, pp. 330–334. arXiv:1303.4194 [cs.CE].
- [9] T. Leinster. 'Rethinking set theory'. In: *arXiv preprint arXiv:1212.6543* (2012).
- [10] E. Maskin. 'The unity of auction theory: Milgrom's master class'. In: *Journal of Economic Literature* 42.4 (2004), pp. 1102–1115.
- [11] T. Nipkow and L. C. Paulson. 'Proof pearl: Defining functions over finite sets'. In: *Theorem Proving in Higher Order Logics*. Springer, 2005, pp. 385–396.
- [12] L. C. Paulson. 'Defining functions on equivalence classes'. In: *ACM Transactions on Computational Logic (TOCL)* 7.4 (2006), pp. 658–675.
- [13] L. C. Paulson. 'Set Theory for Verification: I. From Foundations to Functions'. In: *Journal of Automated Reasoning* 11 (1993), pp. 353–389.
- [14] L. C. Paulson et al. *A Proof Assistant for Higher-Order Logic*. 2013.
- [15] S. G. Simpson. *Subsystems of second order arithmetic*. Vol. 1. Cambridge University Press, 2009.
- [16] S. H. Valentine et al. *AZ Patterns Catalogue II-definitions and laws, v0.1*. 2004.